Jimple: Simplifying Java Bytecode for Analyses and Transformations^�

Raja Vallée-Rai Laurie J. Hendren

Sable Research Group McGill University

fkor,hendreng@sable.mcgill.ca

Abstract

In this paper we present Jimple, a 3-address intermediate representation that has been designed to simplify analysis and transformation of Java bytecode. We motivate the need for a new intermediate representation by illustrating several dif^culties with optimizing the stack-based Java bytecode directly. In general, these dif^culties are due to the fact that bytecode instructions affect an expression stack, and thus have implicit uses and de^nitions of stack locations. We propose Jimple as an alternative representation, in which each statement refers explicitly to the variables it uses. We provide both the de^nition of Jimple and a complete procedure for translating from Java bytecode to Jimple. This de^nition and translation have been implemented using Java, and ^nally we show how this implementation forms the heart of the Sable research projects.

1 Introduction

The Java programming language is gaining a lot of popularity, both as a language for programming small applets on the web, and as a language for developing larger application systems. Although the design of the source language and target bytecode are quite clean, there remains signi^cant challenges in providing Java compilers and run-time environments that can exhibit run-time performance comparable to what is currently possible for applica-

�This research was supported, in part, by NSERC and IBM’s Centre for Advanced Studies (CAS). Java is a trademark of Sun Microsystems Inc.

tions written in C/C++. Improving the performance of Java needs to be attacked at many levels, including: static bytecode optimization, Just-In-Time (JIT) compiler technology and improved run-time environments. In all cases, static program analysis can provide important information. This paper reports on the design and implementation of the Jimple intermediate representation that has been designed to simplify the task of developing static analyses and transformations.

When considering the correct intermediate representation for optimization, one might think that analyzing Java bytecode directly would be a good idea. Certainly the bytecode is not too low-level, since all important program information like types is directly available. Further, since bytecode is the ŞbinaryŤ of the Java world, one would avoid spurious translations to a separate intermediate code. However, Java bytecode is a stack-based intermediate representation where instructions have an implicit effect on an evaluation stack (for example, the bytecode instruction iadd pops two integer values and pushes their sum). Many existing analysis and transformation techniques rely on a more traditional intermediate representation such as 3- address instructions, where each instruction has explicit named operands (for example, an addition would be represented as something like x=a+b). Jimple provides such a 3-address representation.

The remainder of the paper is structured as follows. In Section 2 we explore the dif^culties of analyzing and transforming a stack-based representation in more depth, and we motivate the need for a Jimple-like representation. Section 3 gives the de^nition of Jimple, and Section 4 shows how we translate Java bytecode to Jimple. In doing this translation we concentrate on producing correct Jimple code for any veri^able bytecode (even for bytecode that does not come directly from a Java compiler), and we try to mimimize the number of extra statements and variable names that are introduced. Both the Jimple API and the algorithms to translate Java bytecode to Jimple has been implemented in Java. In Section 5 we give a brief overview of how we are using Jimple with our Sable Java projects at McGill and, ^nally, in section 6 we give some conclusions and a brief description of future work.

2 Motivation

A central theme of this paper is that we prefer to develop analyses and optimizing transformations on Jimple, a 3-address intermediate representation, rather than optimizing and transforming the stackbased bytecode directly. In this section we discuss the bene^ts of stack-based code in general (Section 2.1), and then examine some disadvantages of analyzing and optimizing the stack-based code directly (Section 2.2).

2.1 Benefits of stack-based representations

The stack machine model for the Java Virtual Machine was perhaps a reasonable choice for a few reasons. Stack machine interpreters are relatively easy to implement and this was originally important because the goal was to implement the Java Virtual Machine on as many different platforms as possible. More relevantly, stack-based code tends to be compact and this is essential to allow class ^les to be rapidly downloaded over the Internet. A third justi^cation for this model is that it simpli^es the task of code generation. Since the operand stack can be used to store intermediate results, simple traversals of the code’s abstract syntax tree(AST) suf^ce to generate correct Java bytecode.

There are two good reasons for manipulating Java bytecode directly:

The stack code is immediately available: No transformations are required to get the stack code in this form, as it is the native form found in Java class^les. This is important if execution speed is critical (such as for JIT compilers).

The resultant stack code is �nal: Since the code does not need to be transformed to be stored in the class^les, we have complete control over what gets stored. This is important for obfuscation since many of the obfuscation techniques make heavy use of the stack to confuse decompilers, and a 3-address code representation hides the stack.

In speci^c cases, such as those mentioned above, a stack-based representation is useful. However, in the general case where we optimize class^les of- ~ine, these advantages pale in comparison to the following disadvantages.

2.2 Problems of optimizing stack-based code

Even though there are advantages for choosing a stack-based intermediate representation, there are potential disadvantages with respect to program analysis and optimization. To analyze and transform Java bytecode directly, one is forced to add an extra layer of complexity to deal with the complexities of the stack-based model. Given that it is of critical importance to optimize Java this drawback is very important and must be eliminated to allow the clearest and most ef^cient development of optimizations on Java.

Below, we enumerate some ways in which stackbased Java bytecode is complicated to optimize.

Expressions are not explicit: In 3-address code, expressions are explicit. Usually they only occur in assignments (such as

x=a+b) and branch statements (such as an
if a¡b goto L1). There is a ^xed set of

possible expressions, simplifying analyses by restricting the number of cases to consider. For the purposes of this section, we shall distinguish two classes of Java bytecode instructions: the expression instructions, and action instructions. Expression instructions are those which only produce an effect on the operand stack. Examples of this class are: iload, iadd, imul, pop. Action instructions, on the other hand, produce a side effect, such as modifying a ^eld(putfield), calling a method (invokestatic) or storing into a local variable istore. These instructions have concrete effects, whereas the expression instruction are merely used to build arguments on the stack.

Thus in order to determine the expression being acted upon by an action instruction, you need to parse the expression instructions and reconstruct the expression tree, whereas in Jimple these are readily available. And as the next points illustrate, this reconstruction process is not a trivial problem.

Expressions can be arbitrarily large: In order to determine the expression being computed by expression instructions, the analysis must examine the instructions preceding the action instruction and build an expression tree. For a simple case such as:

iconst 5
iload 0
iadd
istore 1

it is easy to determine that the expression being stored in var1 is 5 + var0. In some cases, such as:

iload 3
iconst 5
iload 6
iload 3
iadd
imul
idiv
istore 0

the expression tree is more complex. In this case it is (var3 + 5) * var6 / var0. Variable expression length is a complication, some analyses such as common subexpression elimination require having simple 3-address code expressions available to be implemented ef^ciently. To use these expression trees in such analyses, they would need to ^rst be simpli^ed to use temporary locals, which the Jimple form provides directly.

Concrete expressions can not always be constructed: Due to the nature of the operand stack, the associated expression instructions for a given action instruction are not necessarily immediately next to it. The following store still stores var0 + 5 in var1, despite the intermingled bytecode instructions which store

var2 * var3 in var4.

                                                iconst 5
                                                iload 0
                                                iadd
                                                iload 2
                                                iload 3
                                                imull
                                                istore 4
                                                istore 1

If a complete sequence of expression instructions reside in a basic block, then it is always possible to recover the computed expression. Since the Java Virtual Machine does not require a zero stack depth across control ~ow junctions, an expression used in a basic block can be partially computed in a different basic block. Consider the following example:

iload 0
iload 2
if˙icmpeq label1
goto label2

label1:
ineg

label2:
istore 1

When computing the possible de^nitions for a variable in a 3-address code intermediate representation, the number of possible de^nitions can not exceed the number of assignments to that variable. This example illustrates that this is not the case with stack code, for a single assignment (istore 0) can yield two different de^nitions (-var0 or var0). By allowing the control ~ow to produce such conditional expressions obviously increases the complexity of analyses such as reaching de^nitions and optimizations such as copy and constant propagation. Instead of considering just assignments, they must consider the origins of expressions and their possible multiplicity. Another way of looking at this problem is that some bytecode expressions, like ineg, are implicitly de^ning a variable, but the name of the variable is not explicit, it is simply some stack location. By transforming to Jimple 3- address code all de^nitions become explicit de^nitions of named variables.

Simple transformations become complicated:

The main reason why stack code complicates analyses and transformations is its piecemeal form. The fact that the expression is split into several pieces and is separated from the action instruction causes almost all the complications, for as a result, you can interleave these instructions with other instructions, and spread them over control ~ow boundaries. Transforming the code in this form is dif^cult because all the separate pieces need to be kept track of and maintained. To illustrate this point, this subsection considers the problems associated with performing dead code elimination.

In 3-address code, eliminating a statement is often accomplished by simply deleting it from a graph or list of statements. In Java bytecode, removing an action instruction is similar, except that you also must remove all the associated expression instructions, in order to avoid accumulating unused arguments on the stack. This sounds relatively simple, but there is a catch: if the set of expression instructions cross a control ~ow boundary, then this may not be possible, because other paths depend on the stack depth to be a certain height. For example:

     iload 0
     iload 1
    / “
iadd imul
istore 5 istore 5
... use(5)
            ...

Despite the fact that on the left hand side the local variable 5 is dead, the iadd and

istore 5 cannot be simply deleted, be-

cause we must ensure the two arguments on the stack are still consumed. The best we can do is replace the two instructions with two

pops. Note that a more sophisticated analy-

ses would hoist the imul, istore 5 sequence to one basic block higher to achieve a better result, but this shows exactly how simple transformations have become complicated. In Jimple this problem does not occur, because a dead assignment can be eliminated without any fear of affecting other areas of the program.

In general, transforming Jimple code is simpler because the fundamental units (instances of 3-address code) are self contained and do not depend on other nearby units for their meaning.

For developing analyses and transformations, it should be clear that working with 3-address code, like Jimple, is much simpler and more ef^cient than dealing with stack code. Thus the need for a correct and ef^cient way of transforming Java bytecode to and from Jimple in to make it pro^table to optimize the code in Jimple.

The remainder of this paper gives a correct transformation from Jimple to Java bytecode. Future work will optimize the transformations, while retaining correctness.

3 Definition of Jimple

The Jimple intermediate representation comes in two ~avors: external and internal. The external representation is a regular ASCII text ^le which represents a class^le in its Jimple form. This is important in order to store Jimple code and provides compiler tools with a new language to target. In general, however, manipulating text ^les to perform optimizations is cumbersome and so an internal representation to use in programs is needed. We supply such an internal representation in the form of a Jimple API (written in Java, of course) and this API enables us to represent and manipulate Jimple code in a straightforward and elegant manner.

Thus the Jimple intermediate representation can be de^ned concretely in two different ways. We present a grammar for the external version in ^gures 1 and 2, and the latest speci^cation of the Jimple API can be found on the web at

http://www.sable.mcgill.ca/jimple.

Both forms have the following essential characteristics:

ffl The code is stackless.

ffl Expressions are restricted to the least number of operands (2 in most cases, such as for arithmetic expressions, but possible more for some method calls), and these operands must either be constants or locals.

ffl All local variables must be explicitly declared and typed.

ffl Some local variables can be assigned initial special roles (such as the containers for the passed parameters, or a caught exception.)

See ^gure 10 for a simple example of a complete Jimple method.

Finally, we would like to note that Jimple’s design was inspired by SIMPLE [HS92], an AST to represent C statements in a simple manner also to facilitate optimizations. Furthermore, our current design of Jimple is in its third generation, and notes on the previous incarnations and their problems can be found in [VR98].

4 Translating Bytecode to Jimple

Generating correct Jimple code from bytecode has two fundamental requirements. The generated code must be stackless, and all local variables must be explicitly typed. Since we are interested in optimizing this Jimple code, however, we also require that code be compact. To illustrate these three requirements, consider the transformation of the following bytecode.

  iload 0 int x, y, z;
  iload 1
  iadd --¿ z = x + y;
  istore 2

Original Java Good Jimple code
  bytecode

In the produced Jimple code, the stack has disappeared, all local variables are typed and the statement is as compact as possible. Thus this is a good of example of the type of code we want to generate. To satisfy these requirements and generate good Jimple code, we propose a ^ve step transformation.

4.1 Five Steps to Jimple

1. Producing Verbose Typeless Jimple: The ^rst step to produce Jimple code is to produce verbose typeless Jimple. This is performed by mapping the operand stack onto a set of untyped Jimple local variables, and transforming the implicit references to the stack operands in the stack code to explicit references to the local variables in Jimple. An illustration:

.numlocals 2 unknown op0, op1;
.maxdepth 2 unknown a, b;

iload 0 op0 = a;
iload 0 op1 = a;
imul op0 = op0 * op1;
istore 1 b = op0;

Java bytecode Jimple code

Note that before translating a given bytecode instruction to its Jimple counterpart, the height of the operand stack must be determined before that bytecode.. This is necessary in order to generate the correct references to the Jimple local variables representing the stack. (This is illustrated in the previous example by noting that the same Java bytecode instruction generates different Jimple bytecode based on its position.) Determining the height of the operand stack is achieved through a simple traversal of the code which models the stack effect of each instruction. This modeling process (as well as the translation process) is tedious to implement due to the sheer number different Java bytecode instructions to handle.

The following table summarizes the effect on the stack and the Jimple code generated for a few Java bytecode instructions.

bytecode stack code
          effect generated

iadd nt=t-1 op(nt)=op(t)+op(t-1)
iload x nt=t+1 op(t) = x
ladd nt=t-2 op(nt)=op(t)+op(t-2)
nop nt=t ¡nothing¿
pop nt=t-1 ¡nothing¿

t is the top index of the stack
nt is the new top index

There are two interesting points to note:

ffl Generated typeless Jimple at this point is self-contained. We could stop after step one and have a valid intermediate representation on which to operate. We are interested, however, in typing the local variables since this is essential for higher level optimizations such as type based alias analysis[DMM98].

ffl Producing correct typeless Jimple is possible because the Java Virtual Machine guarantees that the depth of the operand stack is ^xed for every program point throughout the execution of the program[LY96].

2. Compacting the Code: The previous stage produces correct, but extremely verbose code. A simple Java assignment statement such as

x = a * a is converted to four Jimple in-

structions, as illustrated by the example in step 1. The code can be compacted by simply performing copy/constant propagation and dead code elimination. More precisely, the algorithm we use is:

jimple˙method = local˙declarations identity˙stmts
    stmts exception˙ranges;
local˙declarations = local˙declaration local˙declarations — ;
stmts = label ”:” stmt ”;” stmts —
    stmt ”;” smts — ;
exception˙ranges = exception˙range exception˙ranges — ;
local˙declaration = type name ”;”;
identity˙stmts = identity˙stmt identity˙stmts — ;

stmt = breakpoint˙stmt — assign˙stmt — enter˙monitor˙stmt —
    goto˙stmt — if˙stmt — invoke˙stmt — lookup˙switch˙stmt — nop˙stmt — ret˙stmt —
    return˙stmt — return˙void˙stmt — table˙switch˙stmt — throw˙stmt;
breakpoint˙stmt = ”breakpoint”;
assign˙stmt = variable ”=” rvalue;
identity˙stmt = local ”:=” identity˙value;
enter˙monitor˙stmt = ”entermonitor” immediate;
exit˙monitor˙stmt = ”exitmonitor” immediate;
goto˙stmt = ”goto” label;
if˙stmt = ”if” condition ”then” label;
invoke˙stmt = invoke˙expr;
lookup˙switch˙stmt = ”lookupswitch(” immediate ”) –” cases ” default: goto ” label ”˝”
nop˙stmt = ”nop”;
ret˙stmt = ”ret” local;
return˙stmt = ”return” immediate;
return˙void˙stmt = ”return” ;
table˙switch˙stmt = ”tableswitch(” immediate ”) –” cases ” default: goto ” label ”˝”
throw˙stmt = ”throw” immediate;

cases = case cases — ;
case = ”case ” int˙constant ”: goto ” label;

condition = eq˙expr — ge˙expr — le˙expr — lt˙expr — ne˙expr — gt˙expr;
rvalue = array˙ref — constant — expr — instance˙field˙ref — local —
    next˙next˙stmt˙address — static˙field˙ref;
identity˙value = caught˙exception˙ref — parameter˙ref — this˙ref;
variable = array˙ref — instance˙field˙ref — static˙field˙ref — local;
immediate = constant — local;

expr = binop˙expr — cast˙expr — instance˙of˙expr — invoke˙expr — new˙array˙expr —
    new˙expr — new˙multi˙array˙expr — unop˙expr;
binop˙expr = add˙expr — and˙expr — cmp˙expr — cmpg˙expr — cmpl˙expr — div˙expr —
    eq˙expr — ge˙expr — gt˙expr — le˙expr — lt˙expr — mul˙expr — ne˙expr — or˙expr —
    rem˙expr — shl˙expr — shr˙expr — sub˙expr — ushr˙expr — xor˙expr;
add˙expr = immediate ”+” immediate;
and˙expr = immediate ”&” immediate;
cmp˙expr = immediate ”cmp” immediate;
cmpg˙expr = immediate ”cmpg” immediate;
cmpl˙expr = immediate ”cmpl” immediate;
div˙expr = immediate ”/” immediate;
eq˙expr = immediate ”==” immediate;
ge˙expr = immediate ”¿=” immediate;
gt˙expr = immediate ”¿” immediate;
le˙expr = immediate ”¡=” immediate;
lt˙expr = immediate ”¡” immediate;
mul˙expr = immediate ”*” immediate;

Figure 1: Grammar for external Jimple. (part 1 of 2)

ne˙expr = immediate ”!=” immediate;
or˙expr = immediate ”—” immediate;
rem˙expr = immediate ”%” immediate;
shl˙expr = immediate ”¡¡” immediate;
shr˙expr = immediate ”¿¿” immediate;
sub˙expr = immediate ”-” immediate;
ushr˙expr = immediate ”ushr” immediate;
xor˙expr = immediate ”xor” immediate;
cast˙expr = ”(” type ”)” immediate;

instance˙of˙expr = immediate ”instanceof” ref˙type;
invoke˙expr = interface˙invoke˙expr — special˙invoke˙expr — static˙invoke˙expr —
    virtual˙invoke˙expr;
static˙invoke˙expr = ”staticinvoke” ”[” method˙signature ”](” immediate˙list ”)”;
interface˙invoke˙expr = ”interfaceinvoke” immediate ”.[” + method˙signature ”]”
    ”(” immediate˙list ”)”;
special˙invoke˙expr = ”specialinvoke” immediate ”.[” method˙signature ”]”
    ”(” immediate˙list ”)”;
virtual˙invoke˙expr = ”virtualinvoke” immediate ”.[” method˙signature ”]”
    ”(” immediate˙list ”)”;

immediate˙list = immediate — immediate immediate˙next˙list — ;
immediate˙next˙list = ”,” immediate immediate˙next˙list — ;
new˙array˙expr = ”new” type ”[” immediate ”]”;
new˙expr = ”new” ref˙type ”()”;
new˙multi˙array˙expr = ”new multiarray ” type sized˙dims empty˙dims;

sized˙dims = ”[” immediate ”]” next˙sized˙dims;
next˙sized˙dims = ”[” immediate ”]” next˙sized˙dims — ;
empty˙dims = ”[]” empty˙dims — ;
unop˙expr = length˙expr — neg˙expr;
length˙expr = ”length” immediate;
neg˙expr = ”-” immediate;
array˙ref = immediate ”[” immediate ”]”;
instance˙field˙ref = immediate ”.[” field˙signature ”]”;
static˙field˙ref = ”[” field˙signature ”]”;

method˙signature = identifier ”(” type˙list ”):” type;
field˙signature = identifier ”:” type;
type˙list = type type˙next˙list — ;
type˙next˙list = ”,” type type˙next˙list — ;
caught˙exception˙ref = ”@caughtexception”;
parameter˙ref = ”@parameter” int˙constant;
this˙ref = ”@this”;

label = identifier;
local = identifier;
constant = double˙constant — float˙constant — int˙constant — long˙constant —
    string˙constant — null˙constant;
type = int˙type — long˙type — float˙type — double˙type — ref˙type —
    stmt˙address˙type — void˙type;
exception˙range = ”.catch” ref˙type ”from” label ”to” label ”using” label;

Figure 2: Grammar for external Jimple (part 2 of 2.)

until the code stops changing perform constant and copy propagation perform dead code elimination

Jimple is to be used as an intermediate representation on which to perform optimizations. This step, however, made us develop optimizations right away just in order to produce Jimple. In doing so, we partially tested our framework’s usability, and the results are encouraging. The analyses (reaching de�nitions, using de�nitions, and available copies) and transformations were extremely straightforward to implement. (De^nitions for these basic analyses can be found in [Muc97] and [ASU86]).

Here are some examples of compacting some code produced from step 1:

  Before After

op0 = a
op1 = b x = a + b
op0 = a + b
x = op0

op0 = a call f(a, b)
op1 = b
call f(op0, op1)

Our cleanup algorithm is somewhat time consuming because data ~ow analysis is time consuming. One might argue that no such analyses are needed because the Jimple code could be collapsed into aggregated expressions as it is produced, avoiding all these writes and reads from temporary locals. This proposed algorithm would fail, however, to deal with the special case of expression computations spanning basic blocks. Spanning expressions depend on values to be stored in speci^c stack positions, and so explicit reads and writes to the locals representing these stack positions are required to produce equivalent Jimple. Consider the following example:

iload 0 iload 2
iload 1 iload 3
       “/
       /“
iadd imul
istore 4 istore 4
... ...

In this case it is impossible to produce more compact Jimple statements to represent this code, since the arguments of iadd and imul are variable. Our proposed algorithm, however, correctly produces equivalent typeless Jimple code:

op0 = a op0 = c
op1 = b op1 = d
           “/
           /“
op0=op0+op1 op0=op0*op1
e=op0 e=op0
... ...

Another issue worth considering is the placement of this compaction step. Clearly it could be performed at the end, when the typed Jimple is produced, and possibly yield better results since no further transformations occur. According to our experiments, however, we determined that placing the compaction at the second step yields the best execution speed and that the increase in code size or number of locals is negligeable in comparison (see ^gure 3.) If the size of the code is critical, then the few extra statements produced in the jimpli^cation process could be eliminated in an optimizer based on Jimple anyway. Note that a signi^cant speed-up is achieved for the latter two benchmarks (61% and 64% faster, respectively). This occurs because the algorithms used in subsequent steps execute faster by dealing with a smaller set of statements and local variables.

3. Splitting the Local Variables Having generated typeless compact Jimple code, all that remains is to type it. Unfortunately, typing can not be performed directly on the code at this point. In the example^1

op0 = ”hello world”;
string = op0;
op0 = 1;
unit = op0;

the variable op0 is used as both a String and an int. Obviously one can not assign a single type to this variable without causing type con~icts. This type re-use occurs in Jimple because it is allowed in the Java bytecode. The Java Virtual Machine speci^cation allows local variables and stack positions to be used as different types, as long as it is done consistently so that no type con~icts occur. This

1Note that this code can not actually be generated by step 2 since it is not compact.

baf.Method jimple.StmtGraph DustV2.Play^eld locals stmts time ratio locals stmts time ratio locals stmts time ratio compacting:step 2 103 216 1.00 52 261 1.00 234 1958 1.00 compacting:step 5 107 216 1.17 58 252 1.61 257 1940 1.64

Figure 3: The effect of compacting the code at different steps for three different benchmarks. The counts reported above indicate the ^nal number of locals and statements in the jimpli^ed code. The time ratio compares the execution of time to execution time for the jimpli^cation when the clean-up is in step two.

property is enforced by the type analysis stage of the class^le veri^cation[LY96].

We can transform the preceeding code into typable code by simply renaming the last two references of op0 to op1:

op0 = ”hello world”;
string = op0;
op1 = 1;
unit = op1;

This idea of renaming references can be generalized, so that local variables are prevented from being re-used at all. Avoiding all re-uses would cause a split of x in the following example, even if it is used as the same type.

x = 5; x1=5;
use(x); use(x1);
x = 10; x2=10;
use(x); use(x2);

before after

Although this might seem wasteful, splitting local variables in this manner guarantees, by de^nition, that they are not re-used. Hence, they are not re-used as different types and are now eligible to be typed.

Our intuitive notion of a local variable’s multiple uses is captured by the notion of webs[Muc97]. A web is a set S of references (both de^nitions and uses in the regular sense^2) which is minimal and closed. Closed in the sense that for every de^nition d, the regular uses of d are in S, and for all uses u, the possible de^nitions are also in S. Minimal in the sense that it should not have a strict subset which is closed.

Given that we have the analyses of reaching de�nitions and using de^nitions available, building the set of webs for a program is straightforward:

2Like on the right hand side of an assignment statement, when the variable is read.

for every definition d if d is not in a web W = new web S = new set S.add(d)

iterate over x in S while S is not empty W .add(x) x is def: add all uses of x to S x is use: add all defs of x to S

In plain words, the action of this algorithm is as follows. Take a particular de^nition, and put it in a web. Take all of the uses (in the using de�nitions sense) of this de^nition, and add them to the web. For every use added to the web, add all the possible de^nitions to the web. Iterate these last two sentences until a ^xed point is reached, at which point all of these uses and de^nitions de^ne a particular use of the variable, as we are interested in ^nding.

With these webs at our disposal, the splitting algorithm is simply:

construct the webs of uses and definitions split the locals so that every web has its own associated local

Here is the result of applying this splitting algorithm on an example with multiple uses and de^nitions.

      Before After

x = 1 x = 2 x1 = 1 x1 = 2
       “/ “/
      y=x y = x1
      x = 3 x2 = 3
      y = y + x y = y + x2
      print(x) print(x2)

4. Typing the Local Variables Now that we have Jimple code which can be typed, we can worry about how to type it. A straightforward approach to this problem is to build a set of type restrictions for every local variable, based on how the local variables are used in the code. Since all local variables must be de^ned before they are used (this is guaranteed by [LY96]) we decided to restrict the set of references to consider to only the definitions. In the case of primitive types (e.g.

int, double), the type restrictions are in

fact type equalities. These equalities will always be consistent, because code such as

x = 1 x = 2.5
       “/
      print(x)

can not be generated from the previous step, since the veri^er would reject it before step 1 is reached.

Typing reference local variables is a more interesting problem. An assignment to a reference local variable set the type restriction to be that the left hand side is a superclass of the right hand side. Given a set of these restrictions, a simple solution is the least common superclass. For example, if B and C extend A, the variable x in

x = new B x = new C
          “/
      print(x.toString())

could be directly typed as A, the least common superclass of B and C. Note that collecting type restrictions and assigning types must be performed iteratively. This requirement is illustrated in the following example.

x = new C;
y = null;

label1:
    x = y;
    y = new B;
    goto label1;

The type of x depends on the type of y which is unknown when the ^rst assignment is visited. Only after having determined the type of y to be B, can x = y be visited and x be properly assigned a type. Here is a formal description of the algorithm we have just implicitly described.

assign the type unknown to all variables

until there are no more type changes for every definition d of a local l let rt be the type of the right hand side type(l) = type(l).merge(rt)

definition of merge(u, v):

if u and v are primitives and equal: u if u and v are object references: least common superclass otherwise: type error

Although this algorithm works on most class- ^les we have encountered, it fails in two speci^c situations.

ffl When dealing with nulls. The following code only throws an null pointer exception, but it should still be converted to correct Jimple.

a = null;

print(a[i]);

Our algorithm however types a as a

java.lang.Object since it only

uses information about de^nitions. This causes a type con~ict for the array reference a[i] since a is not an array.

ffl The second problem occurs when dealing with interfaces.

class TypeFail
–
    boolean condition;
    A a; B b; C c; D d;

    void f()
    –
        unknown x;

        if(condition)
            x = c;
        else
            x = d;

        x.a();
    ˝
˝

interface A – void a();˝
interface B – void b();˝
interface C extends A, B –˝
interface D extends A, B –˝

Once again, just looking at the definitions is not suf^cient, because the classes C and D here have two different common superclasses. Choosing the wrong one produces a type con~ict since the method a() cannot be called on an instance of interface B.

These problems with our algorithm indicate that we incorrectly simpli^ed the solution when we decided to only consider de^nitions. Instead, we should take into account every reference to a local variable. In the above examples, using a as an array would require that a be an array, and calling a() on x would require that x be an instance of a subclass of A.

This new algorithm is currently being developed by a member of our research group, and is turning out to be more dif^cult than expected. A future paper will present the complete typing solution when it is reached.^3

5. Packing the Local Variables

At this point, we have compacted and typed our stackless 3-address code, and it seems that we might be done. Unfortunately, we are not, for step 3 has uncompacted some aspects of the code. In addition to having the least number of statements possible, we require that the Jimple code produced we have the smallest number of locals too. And step 3 has potentially resulted in an explosion of locals (see ^gure 11.) The following example illustrates this problem.

x = y + z; x1 = y + z;
print(x); print(x1);
x = a + b; x2 = a + b;
print(x); print(x2);

before splitting after splitting

The ^nal step for our transformation process requires that we reorganize the local variables so that they are re-used whenever possible. This is the opposite problem of step 3. Packing local variables in this manner is similar to the problem of register allocation. The only difference is that we start off with a ^xed set of locals which can be arbitrarily large and we are attempting to minimize their use, as opposed to attempting to map the local variables onto a ^xed set of registers.

The algorithm we use is a tweaked register allocation algorithm based on Chaitin’s graph coloring scheme as described in [ASU86]:

construct an interference graph G for the locals

3In the mean time, you can contact Etienne Gagnon at

gagnon@sable.mcgill.ca for more details.

while there is a local to color

let x be the local with the most interferences remove x from the graph and assign it a color such that its neighbouring locals have a different color, using an already assigned color if possible

assign a local variable to each color, and rename all the references appropriately.

After having performed this algorithm, we ^- nally have transformed our Java bytecode into correct Jimple code which is stackless, typed and compact. Hence our initial requirements for the transformation to this 3-address code intermediate representation are sasti^ed.

The next two sections illustrate the jimpli^cation process with a complete example, and give concrete results on its performance.

4.2 An Example

Here is an example of the entire jimpli^cation process on a method called stepPoly which computes a simple piece-wise de^ned function. The original Java code is in ^gure 4, and the bytecode on which the jimpli^cation process is performed is in ^gure 5. Figures 6 through 10 are snapshots of the code which is internally produced after every step of the transformation.

4.3 Preliminary Results

Figure 11 presents some results of the jimpli- ^cation process on three different benchmarks.

baf.Method and jimple.StmtGraph

come from the Jimple framework, whereas

DustV2.Playfield comes from a numerical

application which simulates electric particle motion.

The numbers presented are the local variable and statements counts of the entire class^le, after a given step of the transformation has been performed. The time percentage indicates the amount of execution time spent for this stage.

There are two interesting facts to note:

ffl After step 2 is performed, the number of statements consistently drops by roughly 40%.

ffl Local packing is essential for good code generation. For the last benchmark, it reduces the local variable count by 81%!

baf.Method jimple.StmtGraph DustV2.Play^eld locals stmts time(%) locals stmts time(%) locals stmts time(%)

step 1: verbose typeless 110 357 34 44 430 29 217 3261 24 step 2: compaction 87 216 34 34 261 31 178 1958 34 step 3: local splitting 134 216 10 181 261 11 1262 1958 11 step 4: type inference 134 216 <1 181 261 <1 1262 1958 <1 step 5: local packing 103 216 15 54 261 24 234 1958 27

Figure 11: Statistics on the local and statement count after each step in jimpli^cation process.

public int stepPoly(int x)
–
    if(x ¡ 0)
    –
        System.out.println(”error”);
        return -1;
    ˝
    else if(x ¡= 5)
        return x * x;
    else
        return x * 5 + 16;
˝

Figure 4: stepPoly in its original Java form.

public int stepPoly(int)
–
    .maxstack 2
    .maxlocals 2

    iload 1;
    iconst 0;
    if˙icmpge label0;

    getstatic [java.lang.System.out:
      java.io.PrintStream];
    sconst ”error”;
    virtualinvoke [java.io.PrintStream.
      println(java.lang.String):void];
    iconst -1;
    ireturn;

 label0:
    iload 1;
    iconst 5;
    if˙icmpgt label1;

    iload 1;
    iload 1;
    imul;
    ireturn;

 label1:
    iload 1;
    iconst 5;
    imul;
    iconst 16;
    iadd;
    ireturn;
˝

Figure 5: stepPoly in its bytecode form.

public int stepPoly(int)
–
    unknown op0, l0, op1, l1;

    l0 := @this;
    l1 := @parameter0;
    op0 = l1;
    op1 = 0;
    if op0 ¿= op1 goto label0;

    op0 = java.lang.System.out;
    op1 = ”error”;
    op0.println(op1);
    op0 = -1;
    return op0;

 label0:
    op0 = l1;
    op1 = 5;
    if op0 ¿ op1 goto label1;

    op0 = l1;
    op1 = l1;
    op0 = op0 * op1;
    return op0;

 label1:
    op0 = l1;
    op1 = 5;
    op0 = op0 * op1;
    op1 = 16;
    op0 = op0 + op1;
    return op0;
˝

Figure 6: stepPoly after step 1. Verbose and typeless.

public int stepPoly(int)
–
    unknown op0, l0, l1;

    l0 := @this;
    l1 := @parameter0;
    if l1 ¿= 0 goto label0;

    op0 = java.lang.System.out;
    op0.println(”error”);
    return -1;

 label0:
    if l1 ¿ 5 goto label1;

    op0 = l1 * l1;
    return op0;

 label1:
    op0 = l1 * 5;
    op0 = op0 + 16;
    return op0;
˝

Figure 7: stepPoly after step 2. It has been compacted.

public int stepPoly(int)
–
    unknown r1, r2, r0, r3, r4, r5;

    r0 := @this;
    r1 := @parameter0;
    if r1 ¿= 0 goto label0;

    r4 = java.lang.System.out;
    r4.println(”error”);
    return -1;

 label0:
    if r1 ¿ 5 goto label1;

    r2 = r1 * r1;
    return r2;

 label1:
    r5 = r1 * 5;
    r3 = r5 + 16;
    return r3;
˝

Figure 8: stepPoly after step 3. The locals have been split to match the webs.

public int stepPoly(int)
–
    java.io.PrintStream r1;
    Example r0;
    int i0, i1, i2, i3;

    r0 := @this;
    i0 := @parameter0;
    if i0 ¿= 0 goto label0;

    r1 = java.lang.System.out;
    r1.println(”error”);
    return -1;

 label0:
    if i0 ¿ 5 goto label1;

    i1 = i0 * i0;
    return i1;

 label1:
    i3 = i0 * 5;
    i2 = i3 + 16;
    return i2;
˝

Figure 9: stepPoly after step 4. It is now typed.

public int stepPoly(int)
–
    java.io.PrintStream r1;
    Example r0;
    int i0;

    r0 := @this;
    i0 := @parameter0;
    if i0 ¿= 0 goto label0;

    r1 = java.lang.System.out;
    r1.println(”error”);
    return -1;

 label0:
    if i0 ¿ 5 goto label1;

    i0 = i0 * i0;
    return i0;

 label1:
    i0 = i0 * 5;
    i0 = i0 + 16;
    return i0;
˝

Figure 10: stepPoly after step 5. The locals have been packed and this code is ready for optimizations.

Figure 12: The Jimple intermediate representation is at the heart of the Sable research projects at McGill. The diamonds indicate internal API representations, whereas the rectangles indicate external ASCII text representations.

5 Current Uses

The Jimple intermediate representation is currently at the heart of the Sable research projects at McGill. As shown in Figure 12, we also have two other representations, Gimp and Baf. Gimp is used when we wish to form large aggregated expressions which can be used in the conversion back to bytecode and in the decompiler. Baf is a direct abstraction of Java bytecode, and is used to shield us from the encoding issues of dealing with Java class^les directly.

Also note that Jimple is the central representation on which we perform analysis and transformation. We are currently working on a wide variety of pointer analyses, side-effect analyses, and type approximation algorithms. The results of these analyses will be used to optimize Jimple itself by applying traditional scalar optimizations like code motion, redundant and dead code elimination, and optimization of virtual method calls. We also feel that it will be useful to encode the analyses results as attributes in the resulting class ^les. In the case of JIT compilers, pre-computed data~ow information can be used to aid in register allocation, the elimination of extra array bounds checks, and the optimization of checks for exceptions[HAKN97]. In the case of ahead-of-time compilers, the attributes can be used instead of re-evaluating the data~ow information (in fact the ahead-of-time compiler would not even need to implement the analysis). Finally, these attributes may also be used to provide hints to a run-time system that can lead to dynamic optimizations.

6 Conclusions and Future Work

In this paper we presented Jimple, a 3-address representation for Java bytecode. We motivated the need for such a representation, and showed how to convert Java bytecode to Jimple.

It should be noted that many other Java translators probably also have to solve similar problems because of the stack-oriented nature of Java bytecode. This is particularly true when the target language is traditional native code, or other representations that do not use an expression stack. Further, other intermediate forms for Java have also been proposed. One such proposal is Slim Binaries which are based on abstract syntax trees that can be compressed into a very dense representation leading to smaller binaries than class ^les [KF97]. The authors argue that Slim Binaries are also well suited to all levels of optimization. The Jimple representation is not aimed at minimizing code size, but is designed for ease of use (and familiarity) for compiler writers who like to work with 3-address type representations.

In designing Jimple we have concentrated on making a clean API that is easy to use for both analysis and transformations. Jimple has undergone several revisions based on the experience of building analysis and transformations, including those analyses and transformations outlined in Section 4. Further, in translating bytecode to Jimple, we have included techniques that minimize extra names and statements that are introduced by a naive translation from bytecode to 3-address code. By making this code publicly available we hope to provide an alternative to optimizing bytecode directly, and to encourage others to build on the Jimple framework.

Our next goals are to provide automatic translation of Jimple (plus analysis information) back to class ^les with attributes. We expect that some additional work will be required to ensure that we generate ef^cient stack code from Jimple. This will involve good generation for aggregated expressions, and effective packing of Jimple variables into bytecode variables and stack locations.

Acknowledgements

We would like to thank Clark Verbrugge for starting the Jimple project and providing us with initial implementations of Jimple and Cof^ which helped us get off the ground. Also, the elegance of the Jimple API’s is due in part to Etienne Gagnon. He advised us on its design and helped us see the light of object oriented programming.

References

[ASU86] Alfred V. Aho, Ravi Sethi, and Jeffrey D. Ullman. Compilers Principles, Techniques and Tools. Addison- Wesley, 1986.

[DMM98] Amer Diwan, Kathryn S. McKinley, and J. Eliot B. Moss. Type-based alias analysis. In Proceedings of the ACM SIGPLAN ’98 Conference on Programming Language Design and Implementation (PLDI ’98), pages 106Ű 115, Montreal, Canada, June 1998.

[HAKN97] Joe Hummel, Ana Azevedo, David Kolson, and Alex Nicolau. Annotating and optimizing the java bytecodes. In Proceedings of the Intenational Workshop of Security and Ef�ciency Aspects of Java (MASCOTS ’97), pages 9Ű10, Eilat, Israel, January 1997.

[HS92] Laurie J. Hendren and Bhama Sridharan. The SIMPLE AST - McCAT compiler. ACAPS Technical Note 36, ACAPS Research Group, McGill University, October 1992.

[KF97] Thomas Kistler and Michael Frannz. A tree-based alternative to java bytecodes. In Proceedings of the Intenational Workshop of Security and Ef- �ciency Aspects of Java (MASCOTS ’97), pages 9Ű10, Eilat, Israel, January 1997.

[LY96] Tim Lindholm and Frank Yellin. The Java Virtual Machine Speci�cation. Addison-Wesley, 1996.

[Muc97] Steven S. Muchnick. Advanced Compiler Design and Implementation. Morgan Kaufmann, 1997.

[VR98] Raja Vallťee-Rai. The Jimple Framework. Sable Technical Report 1, Sable Research Group, McGill University, February 1998.